

System Principles

version 5.2

Typeset in \LaTeX from SGML source using the DOCBUILDER 3.2.2 Document System.

Contents

1	System Principles	1
1.1	System Principles	1
1.1.1	Starting the System	1
1.1.2	Restarting and Stopping the System	2
1.1.3	Command Line Arguments	3
1.1.4	The Boot File	4
1.1.5	Making a Boot File	5
1.1.6	Starting the System with a Boot File	5
1.1.7	Code Loading Strategy	9
1.1.8	Making an Embedded System	10
1.1.9	The Primitive Loader	10
1.1.10	File Types	11
1.2	Creating a First Target System	11
1.2.1	Introduction	11
1.2.2	Creating a Target System	12
1.2.3	Installing a Target System	13
1.2.4	Starting a Target System	13
1.2.5	System Configuration Parameters	14
1.2.6	Differences from the Install Script	14
1.2.7	Listing of target_system.erl	14
	List of Tables	21

Chapter 1

System Principles

1.1 System Principles

This chapter describes the strategies and options which are available to start the Erlang runtime system of Erlang/OTP. This section includes the following topics:

- Starting the system
- Re-starting and stopping the system
- Command line arguments
- The boot file
- Code loading strategies
- Making a boot file
- Starting the system with a boot file
- Code loading strategy
- Making an embedded system
- The primitive loader.

1.1.1 Starting the System

An Erlang runtime system is started with the command:

```
erl [-boot B] [-config F] [-mode M] [-heart]
    [-loader L] [-id Id] [-nodes N1 N2 ... Nn]
    [-pa Dir1 Dir2 ... Dirn] [-pz Dir1 Dir2 ... Dirn]
    [-path Dir1 Dir2 ... Dirn]
    [-AppName Key Value]
    [Other args]
```

- `-boot B` tells the system to use the boot file named `B.boot` to boot the system. This boot file is responsible for the initial loading of the system. If `B` is not supplied it defaults to `start`. When Erlang starts, it searches for the boot file in the current working directory and then in `$ROOT/bin`, where `$ROOT` is the root of the Erlang distribution. If `-loader distributed -nodes N1 N2` was specified, the script is fetched from one of the nodes `N1`, `N2`, ..., otherwise it is fetched by requesting it from the program given in the argument to the `-loader` parameter.

- `-config F` tells the system to use data in the system configuration file `F.config` to override the arguments contained in the application resource files for the set of applications used by this system.
- `-mode M` is the mode in which the system is started. `M` must be either `embedded` or `interactive`. If `-mode M` is omitted, it defaults to `interactive`. In embedded mode all, modules are loaded at system start.
- `-heart` This argument starts an external program which monitors the Erlang node. If the Erlang node hangs, or terminates abnormally, the heart program can restart the Erlang node.
- `-loader L` defines the loader program `L` which fetches code when the system is started. `L` is either the atom `distributed`, or the name of an external program. If `L` is not supplied, it defaults to `efile` which is the normal Erlang filer.
- `-id Id` gives a unique identifier to each Erlang node. If omitted, `Id` defaults to the atom `none`. This flag is not required if the default loader `efile` is used. If the `-sname Name` or `-name Name` parameters are given, `Id` must be the same as `Name`.
- `-nodes N1 N2 ... Nn` must be supplied if `-loader distributed` is specified. `N1, N2, ..., Nn` are Erlang node names from which the system can fetch the modules to be loaded.
- `-pa Dir1 Dir2 ... Dirn` defines a set of directories, `Dir1, Dir2, .. Dirn` which are added to the front of the standard search path which is defined in the start-up script.
- `-pz Dir1 Dir2 ... Dirn` defines a set of directories, `Dir1, Dir2, .. Dirn` which are added to the end of the standard search path which is defined in the start-up script.
- `-path Dir1 Dir2 ... Dirn` defines a set of directories, `Dir1, Dir2, .. Dirn` which replace the standard search path defined in the start-up script.
- `[-AppName Key Value]` overrides the `AppName` application configuration parameter `Key` with `Value`.
- `[Other Args]` are parsed in a standard manner and can be accessed by any application.

The following comments apply to the arguments listed above:

- The default loader is the program `efile`. Through `erl_prim_loader`, it provides a minimal file system interface between Erlang and the local file system in order to load code.
- When `-loader L` is specified, the primitive code loader must know how to retrieve a boot script with name `B.boot`.
- When `-loader distributed -nodes N1 N2 ... Nn` is specified, the boot servers with registered names `boot_server` are assumed to be running on all Erlang nodes `N1, N2, ..., Nn`. If they are not, the system waits for these boot servers to start. Requests are sent to these boot servers to obtain files with names `{Id, Name}` (`Id` is specified in the command line arguments). The boot servers must know how to map these names onto local file names. A simple boot server `erl_boot_server` is provided with the system.
- The boot file with extension `.boot` is created by evaluating the expression `sys_tools:script2boot("File")`. This function converts the script file `File.script` to a boot file `File.boot`.

1.1.2 Restarting and Stopping the System

The system is restarted and stopped with the following commands:

- `init:restart()`. This command restarts the system *inside* the running Erlang node. All applications are taken down smoothly, and all code is unloaded before the system is started again in the same way as it was started initially.

- `init:reboot()`. All applications are taken down smoothly, and all code is unloaded before the Erlang node terminates. The `heart` argument affects the reboot sequence as follows:
 1. If the `-heart` argument was supplied, the `heart` program tries to reboot according to the `HEART_COMMAND` environment variable.
 2. If this variable is not set, `heart` simply writes to `std_out` that it should have rebooted.
 3. If `HEART_COMMAND` is `/usr/sbin/reboot`, the whole machine is rebooted.
- `init:stop()`. All applications are taken down smoothly, and all code is unloaded. If the `-heart` argument was supplied, the `heart` program is terminated before the Erlang node terminates.

1.1.3 Command Line Arguments

When the system has started, application programs can access the values of the command line arguments by calling one of the functions `init:get_argument(Key)`, or `init:get_arguments()`.

Erlang was started by giving a command of the form:

```
erl -flag1 arg1 arg2 -flag2 arg3 ...
```

When the `erl -flag1 ...` command has been issued, Erlang starts by spawning a new process and the system behaves *as if* the function `spawn(init, boot, [Args])` had been evaluated. `Args` is a list of all the command line arguments to `erl`. These are passed as strings. For example, the command `erl -id 123 -loader efile -script "abc" ...` causes the system to behave as if it had evaluated the following function:

```
spawn(init, boot, ["-id", "123", "-loader", "efile",
                  "-script", "\"abc\""]).
```

The first thing `init` does is to call `init:parse_args(Args)` to “normalize” the input arguments. After normalization, the arguments can be accessed as follows:

- `init:get_argument(Flag) -> {ok, [[Arg]]} | error` tries to fetch the argument associated with `Flag`. The return value is either a list of argument lists, or the atom `error`. Flags can have multiple values. If the command line was `erl -p1 a b c -p2 a x -p1 ww zz`:
 - `init:get_argument(p1)` would return:


```
{ok, [{"a", "b", "c"}, {"ww", "zz"}]}
```
 - `init:get_argument(p2)` would return:


```
{ok, [{"a", "x"}]}
```

This is why `get_argument` returns a list of lists, and not just a list.

- `init:get_arguments() -> [{Flag, [Arg]}]` returns *all* the command line arguments. For the command line given above, this would return:


```
[{p1, ["a", "b", "c"]}, {p2, ["a", "x"]}, {p1, ["ww", "zz"]}]
```

Both `get_arguments/0` and `get_argument/1` preserve the argument order of the arguments supplied with the command line.

Note:

Applications should not normally be configured with command line flags, but should use the application environment instead. Refer to *Configuring an Application in the Design Principles* chapter for details.

1.1.4 The Boot File

The boot script is stored in a file with the extension `.script`

A typical boot script file may look as follows:

```
{script, {Name, Vsn},  
 [  
   {progress, loading},  
   {preLoaded, [Mod1, Mod2, ...]},  
   {path, [Dir1, "$ROOT/Dir", ...]},  
   {primLoad, [Mod1, Mod2, ...]},  
   ...  
   {kernel_load_completed},  
   {progress, loaded},  
   {kernelProcess, Name, {Mod, Func, Args}},  
   ...  
   {apply, {Mod, Func, Args}},  
   ...  
   {progress, started}}].
```

The meanings of these terms are as follows:

- `{script, {Name, Vsn}, ...}` defines the script name and version.
- `{progress, Term}` sets the “progress” of the initialization program. The function `init:get_status()` returns the current value of the progress, which is `{InternalStatus, Progress}`.
- `{path, [Dir]}`. `Dir` is a string. This argument sets the load path of the system to `[Dir]`. The load path used to load modules is obtained from the initial load path, which is given in the script file, together with any path flags which were supplied in the command line arguments. The command line arguments modify the path as follows:
 - `-pa Dir1 Dir2 ... Dirn` adds the directories `Dir1, Dir2, ..., Dirn` to the front of the initial load path.
 - `-pz Dir1 Dir2 ... Dirn` adds the directories `Dir1, Dir2, ..., Dirn` to the end of the initial load path.
 - `-path Dir1 Dir2 ... Dirn` defines a set of directories `Dir1, Dir2, ..., Dirn` which replaces the search path given in the script file. Directory names in the path are interpreted as follows:
 - * Directory names starting with `/` are assumed to be absolute path names.
 - * Directory names not starting with `/` are assumed to be relative to the current working directory.
 - * The special `$ROOT` variable can only be used in the script, not as a command line argument. The given directory is relative to the Erlang installation directory.
- `{primLoad, [Mod]}` loads the modules `[Mod]` from the directories specified in `Path`. The script interpreter fetches the appropriate module by calling the function `erl_prim_loader:get_file(Mod)`. A fatal error which terminates the system will occur if the module cannot be located.
- `{kernel_load_completed}` indicates that all modules which *must* be loaded *before* any processes are started are loaded. In interactive mode, all `{primLoad, [Mod]}` commands interpreted after this command are ignored, and these modules are loaded on demand. In embedded mode, `kernel_load_completed` is ignored, and all modules are loaded during system start.

- `{kernelProcess, Name, {Mod, Func, Args}}` starts a “kernel process”. The kernel process `Name` is started by evaluating `apply(Mod, Func, Args)` which is expected to return `{ok, Pid}` or `ignore`. The `init` process monitors the behaviour of `Pid` and terminates the system if `Pid` dies. Kernel processes are key components of the runtime system. Users do not normally add new kernel processes.
- `{apply, {Mod, Func, Args}}`, The `init` process simply evaluates `apply(Mod, Func, Args)`. The system terminates if this results in an error. The boot procedure hangs if this function never returns.

Note:

In the interactive system the code loader provides demand driven code loading, but in the embedded system the code loader loads all the code immediately. The same version of code is used in both cases. The code server calls `init:get_argument(mode)` to find out if it should run in demand mode, or non-demand driven mode.

1.1.5 Making a Boot File

If a boot script is written manually, the `systools:script2boot(File)` function can be used to generate the compiled (binary) form `File.boot` from the `File.script` file. However, it is recommended that the `systools:make_script` function is used in order to create a boot script.

1.1.6 Starting the System with a Boot File

The command `erl -boot File` starts the system with a boot file called `File.boot`. An ASCII version of the boot file can be found in `File.script`.

The boot file is created by evaluating:

```
systools:script2boot(File)
```

Several standard boot files are available. For example, `start.script` starts the system as a plain Erlang runtime system with the `application_controller` and the kernel applications.

`start.script`

The `start.script` is as follows:

```
{script,{ "OTP APN 181 01", "R1A"},
  [{preLoaded,[init,erl_prim_loader]},
   {progress,preloaded},
   {path,["$ROOT/lib/kernel-1.1/ebin",
          "$ROOT/lib/stdlib-1.1/ebin"]},
   {primLoad,[error_handler,
               ets,
               lib,
               lists,
               slave,
               heart,
```

```
    application_controller,  
    application_master,  
    application,  
    auth,  
    c,  
    calendar,  
    code,  
    erlang,  
    erl_distribution,  
    erl_parse,  
    erl_scan,  
    io_lib,  
    io_lib_format,  
    io_lib_fread,  
    io_lib_pretty,  
    error_logger,  
    file,  
    gen,  
    gen_event,  
    gen_server,  
    global,  
    kernel,  
    net_kernel,  
    proc_lib,  
    rpc,  
    supervisor,  
    sys]],  
{kernel_load_completed},  
{progress,kernel_load_completed},  
{primLoad,[group,  
    user,  
    user_drv,  
    kernel_config,  
    net,  
    erl_boot_server,  
    net_adm]}},  
{primLoad,[math,  
    random,  
    ordsets,  
    shell_default,  
    timer,  
    gen_fsm,  
    pg,  
    unix,  
    dict,  
    pool,  
    string,  
    digraph,  
    io,  
    epp,  
    log_mf_h,  
    queue,  
    erl_eval,
```

```

    erl_id_trans,
    shell,
    erl_internal,
    erl_lint,
    error_logger_file_h,
    error_logger_tty_h,
    edlin,
    erl_pp,
    dets,
    regexp,
    supervisor_bridge]],
{progress,modules_loaded},
{kernelProcess,heart,{heart,start,[]}},
{kernelProcess,error_logger,{error_logger,start_link,[]}},
{kernelProcess,application_controller,
    {application_controller,
    start,
    [{application,
    kernel,
    [{description,"ERTS  CXC 138 10"},
    {vsn,"1.1"},
    {modules,
    [{application,1},
    {erlang,1},
    {group,1},
    {rpc,1},
    {application_controller,1},
    {error_handler,1},
    {heart,1},
    {application_master,1},
    {error_logger,1},
    {init,1},
    {user,1},
    {auth,1},
    {kernel,1},
    {user_drv,1},
    {code,1},
    {kernel_config,1},
    {net,1},
    {erl_boot_server,1},
    {erl_prim_loader,1},
    {file,1},
    {net_adm,1},
    {erl_distribution,1},
    {global,1},
    {net_kernel,1}]}]},
    {registered,
    [init,
    erl_prim_loader,
    heart,
    error_logger,
    application_controller,
    kernel_sup,

```

```
        kernel_config,
        net_sup,
        net_kernel,
        auth,
        code_server,
        file_server,
        boot_server,
        global_name_server,
        rex,
        user]],
    {applications,[]},
    {env,
        [{error_logger,tty},
         {os, {unix, 'solaris'}}]}],
    {maxT,infinity},
    {maxP,infinity},
    {mod,{kernel,[]}}]}],
{progress,init_kernel_started},
{apply,{application,load,
        [{application,
         stdlib,
         [{description,"ERTS CXC 138 10"},
          {vsn,"1.1"},
          {modules,
           [{c,1},
            {gen,1},
            {io_lib_format,1},
            {math,1},
            {random,1},
            {sys,1},
            {calendar,1},
            {gen_event,1},
            {io_lib_fread,1},
            {ordsets,1},
            {shell_default,1},
            {timer,1},
            {gen_fsm,1},
            {io_lib_pretty,1},
            {pg,1},
            {slave,1},
            {unix,1},
            {dict,1},
            {gen_server,1},
            {lib,1},
            {pool,1},
            {string,1},
            {digraph,1},
            {io,1},
            {lists,1},
            {proc_lib,1},
            {supervisor,1},
            {epp,1},
            {io_lib,1},
```

```

        {log_mf_h,1},
        {queue,1},
        {erl_eval,1},
        {erl_id_trans,1},
        {shell,1},
        {erl_internal,1},
        {erl_lint,1},
        {error_logger_file_h,1},
        {erl_parse,1},
        {error_logger_tty_h,1},
        {edlin,1},
        {erl_pp,1},
        {ets,1},
        {dets,1},
        {regexp,1},
        {erl_scan,1},
        {supervisor_bridge,1}]],
    {registered,
     [timer_server,
      rsh_starter,
      take_over_monitor,
      pool_master,
      dets]},
    {applications,[kernel]},
    {env,[]},
    {maxT,infinity},
    {maxP,infinity}]}}],
{progress,applications_loaded},
{apply,{application,start_boot,[kernel,permanent]}},
{apply,{application,start_boot,[stdlib,permanent]}},
{apply,{c,erlangrc,[]}},
{progress,started}}].

```

1.1.7 Code Loading Strategy

The code is always loaded relative to the current path and this path is obtained from the value given in the script file, possibly modified by the path manipulation flags in the command line.

This approach allows us to run the system in a number of different ways:

- *Interactive mode.* The system dynamically loads code on demand from the directories specified in the path command. This is the “normal” way to develop code.
- *Embedded mode.* The system loads all its code during system start-up. In special cases, all code can be located in a single directory. We would copy all files to a given directory and create a path to this directory only.
- *Test mode.* Test mode is typically used if we want to run some new test code together with a particular release of the embedded system. We want all the convenience of the interactive system with code loading on demand, and the rigor of the embedded system. In test mode, we run the system with command line arguments such as `-pa " . "`.

1.1.8 Making an Embedded System

When using the the interactive Erlang development environment, it often does not matter if things go wrong at runtime. The main difference with an embedded system is that it is extremely important that things do not go wrong at runtime.

Before building a release which is targeted for an embedded system, we must perform a large number of compile-time checks on the code.

A boot script file can be created with the `systools:make_script` function. This function reads a `.rel` release file and generates the boot script in accordance with the specified applications in the release file. A boot script which is generated this way ensures that all code specified in the application resource files are loaded and that all specified applications are started.

A complete release can be packaged with the `systools:make_tar` function . All application directories and files are packaged according to the release file. The release file and the release upgrade script are also included in the release package.

1.1.9 The Primitive Loader

Unlike the Erlang node, the primitive file loader “knows” how to fetch modules and scripts from its environment.

The interface to the primitive loader is as follows:

- `erl_prim_loader:start(Id, L, Nodes) -> ok | error` starts the primitive loader with the arguments given in the command line.
- `erl_prim_loader:set_path([Dir]) -> ok` sets the path given in the boot file. The value of `[Dir]` comes from the command `{path, [Dir]}` in the start-up script combined with the command line arguments.
- `erl_prim_loader:get_path() -> {ok, Path}` returns the Path used by the primitive loader.
- `erl_prim_loader:get_file(File) -> {ok, FullName, Bin} | error` loads a file from the current path. File is either an absolute file name or just the name of the file, for example `lists.beam`. `FullName` is the name of the file if the load succeeds. `Bin` is the contents of the file as a binary.

Note:

We assume the primitive loader to be running as long as the Erlang node is up and running. In the interactive mode, the code server fetches all code through the loader and the `application_controller` fetches configuration and application files this way.

If an other loader than the one distributed with the system is required, this loader must be implemented by the user as an external port program. The Loader provided by the user must fulfill a protocol defined for the `erl_prim_loader`, and it will be started by the `erl_prim_loader` using the `open_port({spawn, Loader}, [binary])` function call. Refer to the Reference Manual for more information.

1.1.10 File Types

The following file types are defined in Erlang/OTP:

Type	File name/Extension	Description	Manual page which describes the file syntax
module	.erl	Erlang code	-
application	.app	Application resource file	app(4)
release	.rel	Release resource file	rel(4)
script	.script	Start script	script(4)
boot	.boot	Binary boot file	-
config	.config	Configuration file - used to override values in the .app files	config(4)
application upgrade	.appup	Application upgrade	appup(4)
release upgrade script	relup	Release upgrade script	relup(4)

Table 1.1: File Types

1.2 Creating a First Target System

1.2.1 Introduction

Often it is not desirable to use an Erlang/OTP system as is. A developer may create new Erlang/OTP compliant applications for a particular purpose, and several original Erlang/OTP applications may be irrelevant for the purpose in question. Thus, there is a need to be able to create a new system based on a given Erlang/OTP system, where dispensable applications are removed, and a set of new applications that are included in the new system. Documentation and source code is irrelevant and is therefore not included in the new system.

This chapter is about creating such a system, which we call a *target system*.

In the following sections we consider creating target systems with different requirements of functionality:

- a *basic target system* that can be started by calling the ordinary `erl` script,
- a *simple target system* where also code replacement in run-time can be performed, and
- an *embedded target system* where there is also support for logging output from the system to file for later inspection, and where the system can be started automatically at boot time.

We only consider the case when Erlang/OTP is running on a UNIX system.

There is an example Erlang module `target_system.erl` that contains functions for creating and installing a target system. That module is used in the examples below. The source code of the module is listed at the end of this chapter.

1.2.2 Creating a Target System

It is assumed that you have a working Erlang/OTP system (which you can start with the command `erl`).

Step 1. First create a `.rel` file (see the `rel(4)` manual page for details) that specifies the `erts` version and lists all applications that should be included in the new basic target system. An example is the following `mysystem.rel` file:

```
%% mysystem.rel
{release,
 { "MYSYSTEM", "FIRST"},
 {erts, "5.1"},
 [{kernel, "2.7"},
 {stdlib, "1.10"},
 {sas1, "1.9.3"},
 {pea, "1.0"}]}
```

The listed applications are not only original Erlang/OTP applications but possibly also new applications that you have written yourself (here exemplified by the application `pea`).

Step 2. From the directory where the `mysystem.rel` and `plain.rel` files reside, start the Erlang/OTP system:

```
erl -pa /home/user/target_system/myapps/pea-1.0/ebin
```

where also the path to the `pea-1.0` `ebin` directory is provided.

Step 2. Now create the target system:

```
1> target_system:create("mysystem").
```

The `target_system:create/1` function does the following:

1. Reads the `mysystem.rel` file, and creates a new file `plain.rel` which is identical to former, except that it only lists the `kernel` and `stdlib` applications.
2. From the `mysystem.rel` and `plain.rel` files creates the files `mysystem.script`, `mysystem.boot`, `plain.script`, and `plain.boot` through a call to `systools:make_script/2`.
3. Creates the file `mysystem.tar.gz` by a call to `systools:make_tar/2`. That file has the following contents:

```
erts-5.1/bin/
releases/FIRST/start.boot
releases/mysystem.rel
lib/kernel-2.7/
lib/stdlib-1.10/
lib/sasl-1.9.3/
lib/pea-1.0/
```

The file `releases/FIRST/start.boot` is a copy of our `mysystem.boot`, and a copy of the original `mysystem.rel` has been put in the `releases` directory.

4. Creates the temporary directory `tmp` and extracts the tar file `mysystem.tar.gz` into that directory.
5. Deletes the `erl` and `start` files from `tmp/erts-5.1/bin`. XXX Why.
6. Creates the directory `tmp/bin`.
7. Copies the previously creates file `plain.boot` to `tmp/bin/start.boot`.

8. Copies the files `epmd`, `run_erl`, and `to_erl` from the directory `tmp/erts-5.1/bin` to the directory `tmp/bin`.
9. Creates the file `tmp/releases/start_erl.data` with the contents "5.1 FIRST".
10. Recreates the file `mysystem.tar.gz` from the directories in the directory `tmp`, and removes `tmp`.

1.2.3 Installing a Target System

Step 3. Install the created target system in a suitable directory.

```
3> target_system:install("mysystem", "/usr/local/erl-target").
```

The function `target_system:install/2` does the following:

1. Extracts the tar file `mysystem.tar.gz` into the target directory `/usr/local/erl-target`.
2. In the target directory reads the file `releases/start_erl.data` in order to find the Erlang runtime system version ("5.1").
3. Substitutes `%FINAL_ROOTDIR%` and `%EMU%` for `/usr/local/erl-target` and `beam`, respectively, in the files `erl.src`, `start.src`, and `start_erl.src` of the target `erts-5.1/bin` directory, and puts the resulting files `erl`, `start`, and `run_erl` in the target `bin` directory.
4. Finally the target `releases/RELEASES` file is created from data in the `releases/mysystem.rel` file.

1.2.4 Starting a Target System

Now we have a target system that can be started in various ways.

We start it as a *basic target system* by invoking

```
/usr/local/erl-target/erl
```

where only the `kernel` and `stdlib` applications are started, i.e. the system is started as an ordinary development system. There are only two files needed for all this to work: `bin/erl` file (obtained from `erts-5.1/bin/erl.src`) and the `bin/start.boot` file (a copy of `plain.boot`).

We can also start a distributed system (requires `bin/epmd`).

To start all applications specified in the original `mysystem.rel` file, use the `-boot` flag as follows:

```
/usr/local/erl-target/erl -boot /usr/local/erl-target/releases/FIRST/start
```

We start a *simple target system* as above. The only difference is that also the file `releases/RELEASES` is present for code replacement in run-time to work.

To start an *embedded target system* the shell script `bin/start` is used. That shell script calls `bin/run_erl`, which in turn calls `bin/start_erl` (roughly, `start_erl` is an embedded variant of `erl`).

The shell script `start` is only an example. You should edit it to suite your needs. Typically it is executed when the UNIX system boots.

`run_erl` is a wrapper that provides logging of output from the run-time system to file. It also provides a simple mechanism for attaching to the Erlang shell (`to_erl`).

`start_erl` requires the root directory ("`/usr/local/erl-target`"), the releases directory ("`/usr/local/erl-target/releases`"), and the location of the `start_erl.data` file. It reads the

run-time system version ("5.1") and release version ("FIRST") from the `start_erl.data` file, starts the run-time system of the version found, and provides `-boot` flag specifying the boot file of the release version found ("`releases/FIRST/start.boot`").

`start_erl` also assumes that there is `sys.config` in release version directory ("`releases/FIRST/sys.config`"). That is the topic of the next section (see below).

The `start_erl` shell script should normally not be altered by the user.

1.2.5 System Configuration Parameters

As was pointed out above `start_erl` requires a `sys.config` in the release version directory ("`releases/FIRST/sys.config`"). If there is no such a file, the system start will fail. Hence such a file has to added as well.

If you have system configuration data that are neither file location dependent nor site dependent, it may be convenient to create the `sys.config` early, so that it becomes a part of the target system tar file created by `target_system:create/1`. In fact, if you create, in the current directory, not only the `mysystem.rel` file, but also a `sys.config` file, that latter file will be tacitly put in the appropriate directory.

1.2.6 Differences from the Install Script

The above `install/2` procedure differs somewhat from that of the ordinary `Install` shell script. In fact, `create/1` makes the release package as complete as possible, and leave to the `install/2` procedure to finish by only considering location dependent files.

1.2.7 Listing of `target_system.erl`

```
-module(target_system).
-include_lib("kernel/include/file.hrl").
-export([create/1, install/2]).
-define(BUFSIZE, 8192).

%% Note: RelFileName below is the *stem* without trailing .rel,
%% .script etc.
%%

%% create(RelFileName)
%%
create(RelFileName) ->
    RelFile = RelFileName ++ ".rel",
    io:fwrite("Reading file: \"~s\" ...~n", [RelFile]),
    {ok, [RelSpec]} = file:consult(RelFile),
    io:fwrite("Creating file: \"~s\" from \"~s\" ...~n",
              ["plain.rel", RelFile]),
    {release,
     {RelName, RelVsn},
     {erts, ErtsVsn},
     AppVsns} = RelSpec,
    PlainRelSpec = {release,
                    {RelName, RelVsn},
                    {erts, ErtsVsn},
```

```

        lists:filter(fun({kernel, _}) ->
                        true;
                      ({stdlib, _}) ->
                        true;
                      (_) ->
                        false
                    end, AppVsns)
    },
    {ok, Fd} = file:open("plain.rel", [write]),
    io:fwrite(Fd, "p.~n", [PlainRelSpec]),
    file:close(Fd),

    io:fwrite("Making \"plain.script\" and \"plain.boot\" files ...~n",
    make_script("plain"),

    io:fwrite("Making \"~s.script\" and \"~s.boot\" files ...~n",
              [RelFileName, RelFileName]),
    make_script(RelFileName),

    TarFileName = io_lib:fwrite("~s.tar.gz", [RelFileName]),
    io:fwrite("Creating tar file \"~s\" ...~n", [TarFileName]),
    make_tar(RelFileName),

    io:fwrite("Creating directory \"tmp\" ...~n",
    file:make_dir("tmp"),

    io:fwrite("Extracting \"~s\" into directory \"tmp\" ...~n", [TarFileName]),
    extract_tar(TarFileName, "tmp"),

    TmpBinDir = filename:join(["tmp", "bin"]),
    ErtsBinDir = filename:join(["tmp", "erts-" ++ ErtsVsn, "bin"]),
    io:fwrite("Deleting \"erl\" and \"start\" in directory \"~s\" ...~n",
              [ErtsBinDir]),
    file:delete(filename:join([ErtsBinDir, "erl"])),
    file:delete(filename:join([ErtsBinDir, "start"])),

    io:fwrite("Creating temporary directory \"~s\" ...~n", [TmpBinDir]),
    file:make_dir(TmpBinDir),

    io:fwrite("Copying file \"plain.boot\" to \"~s\" ...~n",
              [filename:join([TmpBinDir, "start.boot"])]),
    copy_file("plain.boot", filename:join([TmpBinDir, "start.boot"])),

    io:fwrite("Copying files \"epmd\", \"run_erl\" and \"to_erl\" from \"n\"
              \"~s\" to \"~s\" ...~n",
              [ErtsBinDir, TmpBinDir]),
    copy_file(filename:join([ErtsBinDir, "epmd"]),
              filename:join([TmpBinDir, "epmd"]), [preserve]),
    copy_file(filename:join([ErtsBinDir, "run_erl"]),
              filename:join([TmpBinDir, "run_erl"]), [preserve]),
    copy_file(filename:join([ErtsBinDir, "to_erl"]),
              filename:join([TmpBinDir, "to_erl"]), [preserve]),

```

```
StartErlDataFile = filename:join(["tmp", "releases", "start_erl.data"]),
io:fwrite("Creating \"~s\" ...~n", [StartErlDataFile]),
StartErlData = io_lib:fwrite("~s ~s~n", [Ertsvsn, RelVsn]),
write_file(StartErlDataFile, StartErlData),

io:fwrite("Recreating tar file \"~s\" from contents in directory \"
    \"tmp\" ...~n", [TarFileName]),
{ok, Tar} = erl_tar:open(TarFileName, [write, compressed]),
{ok, Cwd} = file:get_cwd(),
file:set_cwd("tmp"),
erl_tar:add(Tar, "bin", []),
erl_tar:add(Tar, "erts-" ++ Ertsvsn, []),
erl_tar:add(Tar, "releases", []),
erl_tar:add(Tar, "lib", []),
erl_tar:close(Tar),
file:set_cwd(Cwd),
io:fwrite("Removing directory \"tmp\" ...~n"),
remove_dir_tree("tmp"),
ok.

install(RelFileName, RootDir) ->
    TarFile = RelFileName ++ ".tar.gz",
    io:fwrite("Extracting ~s ...~n", [TarFile]),
    extract_tar(TarFile, RootDir),
    StartErlDataFile = filename:join([RootDir, "releases", "start_erl.data"]),
    {ok, StartErlData} = read_txt_file(StartErlDataFile),
    [ErlVsn, RelVsn|_] = string:tokens(StartErlData, " \n"),
    Ertsvsn = filename:join([RootDir, "erts-" ++ ErlVsn, "bin"]),
    BinDir = filename:join([RootDir, "bin"]),
    io:fwrite("Substituting in erl.src, start.src and start_erl.src to~n"
        "form erl, start and start_erl ...~n"),
    subst_src_scripts(["erl", "start", "start_erl"], Ertsvsn, BinDir,
        [{"FINAL_ROOTDIR", RootDir}, {"EMU", "beam"}],
        [preserve]),
    io:fwrite("Creating the RELEASES file ...~n"),
    create_RELEASES(RootDir,
        filename:join([RootDir, "releases", RelFileName])).

%% LOCALS

%% make_script(RelFileName)
%%
make_script(RelFileName) ->
    Opts = [no_module_tests],
    systools:make_script(RelFileName, Opts).

%% make_tar(RelFileName)
%%
make_tar(RelFileName) ->
    RootDir = code:root_dir(),
    systools:make_tar(RelFileName, [{erts, RootDir}]).
```

```

%% extract_tar(TarFile, DestDir)
%%
extract_tar(TarFile, DestDir) ->
    erl_tar:extract(TarFile, [{cwd, DestDir}, compressed]).

create_RELEASES(DestDir, RelFileName) ->
    release_handler:create_RELEASES(DestDir, RelFileName ++ ".rel").

subst_src_scripts(Scripts, SrcDir, DestDir, Vars, Opts) ->
    lists:foreach(fun(Script) ->
        subst_src_script(Script, SrcDir, DestDir,
            Vars, Opts)
    end, Scripts).

subst_src_script(Script, SrcDir, DestDir, Vars, Opts) ->
    subst_file(filename:join([SrcDir, Script ++ ".src"]),
        filename:join([DestDir, Script]),
        Vars, Opts).

subst_file(Src, Dest, Vars, Opts) ->
    {ok, Conts} = read_txt_file(Src),
    NConts = subst(Conts, Vars),
    write_file(Dest, NConts),
    case lists:member(preserve, Opts) of
        true ->
            {ok, FileInfo} = file:read_file_info(Src),
            file:write_file_info(Dest, FileInfo);
        false ->
            ok
    end.

%% subst(Str, Vars)
%% Vars = [{Var, Val}]
%% Var = Val = string()
%% Substitute all occurrences of %Var% for Val in Str, using the list
%% of variables in Vars.
%%
subst(Str, Vars) ->
    subst(Str, Vars, []).

subst([$%, C| Rest], Vars, Result) when $A =< C, C =< $Z ->
    subst_var([C| Rest], Vars, Result, []);
subst([$%, C| Rest], Vars, Result) when $a =< C, C =< $z ->
    subst_var([C| Rest], Vars, Result, []);
subst([$%, C| Rest], Vars, Result) when C == $_ ->
    subst_var([C| Rest], Vars, Result, []);
subst([C| Rest], Vars, Result) ->
    subst(Rest, Vars, [C| Result]);
subst([], _Vars, Result) ->
    lists:reverse(Result).

subst_var([$%| Rest], Vars, Result, VarAcc) ->
    Key = lists:reverse(VarAcc),

```

```
case lists:keysearch(Key, 1, Vars) of
  {value, {Key, Value}} ->
    subst(Rest, Vars, lists:reverse(Value, Result));
  false ->
    subst(Rest, Vars, [${VarAcc} ++ [${Result}]]);
end;
subst_var([C| Rest], Vars, Result, VarAcc) ->
  subst_var(Rest, Vars, Result, [C| VarAcc]);
subst_var([], Vars, Result, VarAcc) ->
  subst([], Vars, [VarAcc ++ [${Result}]]).

copy_file(Src, Dest) ->
  copy_file(Src, Dest, []).

copy_file(Src, Dest, Opts) ->
  {ok, InFd} = file:rawopen(Src, {binary, read}),
  {ok, OutFd} = file:rawopen(Dest, {binary, write}),
  do_copy_file(InFd, OutFd),
  file:close(InFd),
  file:close(OutFd),
  case lists:member(preserve, Opts) of
    true ->
      {ok, FileInfo} = file:read_file_info(Src),
      file:write_file_info(Dest, FileInfo);
    false ->
      ok
  end.

do_copy_file(InFd, OutFd) ->
  case file:read(InFd, ?BUFSIZE) of
    {ok, Bin} ->
      file:write(OutFd, Bin),
      do_copy_file(InFd, OutFd);
    eof ->
      ok
  end.

write_file(FName, Conts) ->
  {ok, Fd} = file:open(FName, [write]),
  file:write(Fd, Conts),
  file:close(Fd).

read_txt_file(File) ->
  {ok, Bin} = file:read_file(File),
  {ok, binary_to_list(Bin)}.

remove_dir_tree(Dir) ->
  remove_all_files(".", [Dir]).

remove_all_files(Dir, Files) ->
  lists:foreach(fun(File) ->
    FilePath = filename:join([Dir, File]),
    {ok, FileInfo} = file:read_file_info(FilePath),
```

```
case FileInfo#file_info.type of
  directory ->
    {ok, DirFiles} = file:list_dir(FilePath),
    remove_all_files(FilePath, DirFiles),
    file:del_dir(FilePath);
  _ ->
    file:delete(FilePath)
end
end, Files).
```


List of Tables

1.1 File Types 11